
Modifying User Accounts using Visual Basic Script 
 

You can use TRCLI.exe (TRCLILT.exe for LT Edition) and a Visual Basic Script to batch modify user accounts in a 

CSV text file. Assume that a CSV text file named Users.csv contains following lines and you would like to update 

user passwords with new values in this CSV file; 

 

User-1, New-Pass-1 

User-2,New-Pass-2 

: 

User-n,New-Pass-n 
 

In order to update an attribute in a user profile you need to delete it from user attribute and then add it with the new value. 

You can use following Visual Basic Script to modify user accounts listed in Users.csv.; 
 

Dim username, csvValues, strArgs 

 

Set strArgs  = WScript.Arguments 

 

If strArgs.length < 1 then 

 WScript.Echo "Usage : TR-CSV-Modify [CSV File]" 

 WScript.quit 

End If 

 

Set WshShell = WScript.CreateObject ("WScript.Shell") 

Set objFSO   = CreateObject("Scripting.FileSystemObject") 

Set objFile  = objFSO.OpenTextFile(objFSO.GetAbsolutePathName("") & "\" & Trim(strArgs(0))) 

Set objRegEx = CreateObject("VBScript.RegExp") 

 

objRegEx.Pattern = "^(.+),(.+)$" 

 

Do Until objFile.AtEndOfStream 

  strLine = objFile.ReadLine 

  Set colMatches = objRegEx.Execute(strLine)   

 

  If colMatches.Count > 0 Then 

    csvValues = Split(strLine, ",") 

    username  = Trim(csvValues(0)) 

 

    ' Following line runs trcli.exe and remove User-Password attribute in a user profile. 

    ' Use trclilt for LT edition. 

 

    strRun = "%comspec% /c trcli -m " & username & " ""ietf|2"" check" 

    WshShell.Run strRun, 0, True 

 

    ' This line adds new password to the user profile. 

 

    strRun = "%comspec% /c trcli -a " & username & " ""ietf|2"" " & Trim(csvValues(1)) & " check" 

    WshShell.Run strRun, 0, True 

  End If 

Loop 

 

objFile.Close 

 

You can customize this script to modify more attributes in user profiles.  


